# Maintainer: Christian Hesse <mail@eworm.de>
# Contributor: andreas_baumann <abaumann@yahoo.com>
# Contributor: zhuqin <zhuqin83@gmail.com>

pkgname=gengetopt
pkgver=2.23
pkgrel=2
pkgdesc='A tool to write command line option parsing code for C programs'
arch=('loong64' 'x86_64')
url='https://www.gnu.org/software/gengetopt/gengetopt.html'
license=('GPL')
options=('!docs' '!makeflags')
depends=('gcc-libs')
validpgpkeys=('CD7AB2B200F374043F92E5D42485C3A5CF0FF62F') # Gray Wolf <wolf@wolfsden.cz>
source=("https://ftp.gnu.org/gnu/gengetopt/${pkgname}-${pkgver}.tar.xz"{,.sig})
sha256sums=('b941aec9011864978dd7fdeb052b1943535824169d2aa2b0e7eae9ab807584ac'
            'SKIP')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	./configure --prefix=/usr
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	make DESTDIR="${pkgdir}" install
}

